Multilevel Structural Equation Modeling for Cross-National Comparative Research

Bart Meuleman

Special Issue 2019 - Online Appendices

Appendix 1: Mplus syntax for the MCFA example (see Table 1)

```
TITLE: MCFA model for Euroscepticism - isomorphism;

DATA: file is MSEMrepl_evs.txt;

VARIABLE:
 names are age country fearsoc fearcult fearpay fearpow fearjobs
 gender educat1 educat2 inccat1 inccat2 inccat3 inccat9
 jobcat2 jobcat3 jobcat4 jobcat5 etnocen state
 net_benefits_gdp transfers_net migration_eu_1000;

 missing are all (999);

 usevariables = fearsoc fearcult fearpay fearpow fearjobs;

 cluster is country;


ANALYSIS:

 type = twolevel;


MODEL:

!Note: the letters (a)-(e) are used to impose cross-level equality constraints.
 %within%
 fearw by
 fearsoc (a)
 fearcult (b)
 fearpay (c)
 fearpow (d)
 fearjobs (e);
 fearjobs with fearcult;

 %between%
 fearb by
 fearsoc (a)
 fearcult (b)
 fearpay (c)
 fearpow (d)
 fearjobs (e);


OUTPUT:

 sampstat STDYX modindices (5) residual;
```

Appendix 2: Mplus syntax for the MSEM example (see Table 2)

```
TITLE: MSEM for Euroscepticism;

DATA: file is MSEMrepl_evs.txt;

VARIABLE:

 names are age country fearsoc fearcult fearpay fearpow fearjobs
 gender educat1 educat2 inccat1 inccat2 inccat3 inccat9
 jobcat2 jobcat3 jobcat4 jobcat5 etnocen state
 net_benefits_gdp transfers_net migration_eu_1000;

 missing are all (999);

 usevariables = fearsoc fearcult fearpay fearpow fearjobs
 age gender educat1 educat2 inccat1 inccat2 inccat3 inccat9
 jobcat2 jobcat3 jobcat4 jobcat5
 state etnocen
 net_benefits_gdp transfers_net migration_eu_1000;


!The following variables are defined as within only;
 WITHIN = age gender educat1 educat2
 inccat1 inccat2 inccat3 inccat9
 jobcat2 jobcat3 jobcat4 jobcat5
 state etnocen;

!The following variables are defined as between only;
 BETWEEN = net_benefits_gdp transfers_net migration_eu_1000;

 cluster is country;


ANALYSIS:

 type = twolevel;


MODEL:

 %within%
 fearw by
 fearsoc (a)
 fearcult (b)
 fearpay (c)
 fearpow (d)
 fearjobs (e);
 fearjobs with fearcult;

 fearw on age gender educat1 educat2
 inccat1 inccat2 inccat3 inccat9
 jobcat2 jobcat3 jobcat4 jobcat5
 state etnocen;

 %between%
 fearb by
 fearsoc (a)
 fearcult (b)
 fearpay (c)
 fearpow (d)
 fearjobs (e);

 fearb on net_benefits_gdp transfers_net migration_eu_1000;


OUTPUT:
 sampstat STDYX tech1 tech4 tech8 cinterval;
```

Appendix 3: Mplus syntax for the Monte Carlo simulation study (see Table 3)

```
TITLE: Simulation MSEM for Euroscepticism - Bayesian estimation;

MONTECARLO:

 names are age fearsoc fearcult fearpay fearpow fearjobs
 gender educat1 educat2 inccat1 inccat2 inccat3 inccat9
 jobcat2 jobcat3 jobcat4 jobcat5 etnocen state
 net_benefits_gdp transfers_net migration_eu_1000;

 nobservations = 28000;
 nreps = 2;
 ncsizes = 1;
 csizes =  28 (1000);

 WITHIN = age gender educat1 educat2
 inccat1 inccat2 inccat3 inccat9
 jobcat2 jobcat3 jobcat4 jobcat5
 state etnocen;

 BETWEEN =
 net_benefits_gdp transfers_net migration_eu_1000


MODEL POPULATION:

 %within%
 fearw by
 fearsoc@1.000
 fearcult*1.110
 fearpay*0.982
 fearpow*1.079
 fearjobs*1.049;

 fearw*3.488;
 fearsoc*4.247; fearcult*3.718; fearpay*3.368;
 fearpow*3.634; fearjobs*3.318;

 fearjobs with fearcult*-0.837;

 fearw on age*-0.001 gender*0.277 educat1*0.583 educat2*0.477
 inccat1*0.358 inccat2*0.353 inccat3*0.175 inccat9*0.251
 jobcat2*-0.107 jobcat3*-0.328 jobcat4*0.068 jobcat5*-0.073
 state*0.073 etnocen*0.360;

 age*318.956; gender*0.247; educat1*0.219; educat2*0.248; inccat1*0.160;
 inccat2*0.147; inccat3*0.154; inccat9*0.182; jobcat2*0.189; jobcat3*0.055;
 jobcat4*0.068; jobcat5*0.077; state*6.783; etnocen*4.953;


 %between%
 fearb by
 fearsoc@1.000
 fearcult*1.110
 fearpay*0.982
 fearpow*1.079
 fearjobs*1.049;

 fearb*0.280;
 fearsoc*0.222; fearcult*0.043; fearpay*0.143;
 fearpow*0.078; fearjobs*0.326;

 fearb on
 net_benefits_gdp*0.069
 transfers_net*0.073
 migration_eu_1000*0.000;

 net_benefits_gdp*17.580;
 transfers_net*0.819;
 migration_eu_1000*4721.482;


 ANALYSIS:

 type = twolevel;
!For Bayesian estimation, add the following lines.
```

```
!estimator is bayes;
!processors is 2;
!bconvergence = 0.01;


MODEL:

%within%
fearw by
fearsoc@1.000 (a)
fearcult*1.110 (b)
fearpay*0.982 (c)
fearpow*1.079 (d)
fearjobs*1.049 (e);

fearw*3.488;
fearsoc*4.247; fearcult*3.718; fearpay*3.368;
fearpow*3.634; fearjobs*3.318;

fearjobs with fearcult*-0.837;

fearw on age*-0.001 gender*0.277 educat1*0.583 educat2*0.477
inccat1*0.358 inccat2*0.353 inccat3*0.175 inccat9*0.251
jobcat2*-0.107 jobcat3*-0.328 jobcat4*0.068 jobcat5*-0.073
state*0.073 etnocen*0.360;

%between%
fearb by
fearsoc@1.000 (a)
fearcult*1.110 (b)
fearpay*0.982 (c)
fearpow*1.079 (d)
fearjobs*1.049 (e);

fearb*0.280;
fearsoc*0.222; fearcult*0.043; fearpay*0.143;
fearpow*0.078; fearjobs*0.326;

fearb on
net_benefits_gdp*0.069
transfers_net*0.073
migration_eu_1000*0.000;


OUTPUT:
sampstat STDYX tech1 cinterval;
```